Static Moment Method Solution for a Microstrip in an Infinite Dielectric

Keith W.Whites

Last Modified on January 29, 2004

- The microstrip substrate and the infinite space above it are assumed to be the same material having relative permittivity "ϵrbackgnd." Using pulse expansion and point matching for the MM solution with "numcells" segments of width "Δ" uniformly distributed across the infinitely-thin strip, which has a width-over-separation ratio "Woverd." The voltage of the strip conductor is "Vapplied" volts with respect to the ground plane.

- Revisions:
  →10/1/03:First completed.

```
<< Graphics`Graphics`
```

Enter the geometrical parameters and the applied voltage.

```
ClearAll[Woverd, ϵrbackgnd, Vapplied, d, width, Δ,
 numcells, c0, ϵ0, Δx, m, n, Zmn, ϵr, Zmatrix, Vvector, α, α0,
 CPUL, C0PUL, Z0, ϵreff, ϵe, Wod, Z0approx, plot1, plot2]

Woverd := 5. ;
ϵrbackgnd = 1.  ;
Vapplied = 1.  ;
```

- **Compute the moment method solution. "α" and "α0" are the line charge density coefficients when the background relative permittivity is ϵrbackgnd and 1, respectively. Without loss of generality, assume a spacing of d=1 m between the microstrip and the ground plane and compute width for the chosen Woverd.**

```
d = 1. ;
 width := Woverd * d ;
 Δ := width / numcells ;

 c0 = 2.998 * 10^8 ;
 ϵ0 = 8.854 * 10^-12 ;

 Δx[m_, n_] := Δ * (m - n)
 Zmn[m_, n_, ϵr_] :=
  -1 / (4 * Pi * ϵr * ϵ0) * ((Δx[m, n] + Δ / 2) * Log[(Δx[m, n] + Δ / 2)^2 /
          ((Δx[m, n] + Δ / 2)^2 + 4 * d^2)] - (Δx[m, n] - Δ / 2) *
        Log[(Δx[m, n] - Δ / 2)^2 / ((Δx[m, n] - Δ / 2)^2 + 4 * d^2)] -
      4 * d * (ArcTan[2 * d, Δx[m, n] + Δ / 2] -
         ArcTan[2 * d, Δx[m, n] - Δ / 2])) /; m ≠ n
 Zmn[m_, n_, ϵr_] := Δ / (2 * Pi * ϵr * ϵ0) * (1 - Log[Δ / 2]) +
   1 / (4 * Pi * ϵr * ϵ0) *
    (Δ * Log[Δ^2 / 4 + 4 * d^2] - 2 * Δ + 8 * d * ArcTan[Δ / (4 * d)])

 Zmatrix[ϵr_] := Table[Zmn[m, n, ϵr], {m, numcells}, {n, numcells} ]
 Vvector := Table[Vapplied, {numcells}] ;

 α := LinearSolve[Zmatrix[ϵrbackgnd], Vvector]
 α0 := LinearSolve[Zmatrix[1], Vvector]
```

- **Choose the number of pulse basis functions "numcells"then compute the capacitance per unit length "CPUL" assuming a background relative permittivity "ϵrbackgnd." The characteristic impedance of the microstrip "Z0" and the effective relative permittivity "ϵreff" of a TEM wave propagating on this microstrip are computed from both CPUL as well as "C0PUL," which is the capacitance per unit length of the microstrip with a background relative permittivity equal to 1.**

```
numcells := 30

CPUL := Total[α] * Δ
C0PUL := Total[α0] * Δ

Z0 := (c0 * Sqrt[CPUL * C0PUL])^-1
ϵreff := CPUL / C0PUL

N[Woverd]
N[Z0]
N[ϵreff]
```

```
5.

49.7729

1.
```

- **Plot the amplitudes of the line charge density coefficients α. It can be shown theoretically that the line charge density should approach infinity at the edges of the strip. This is called the "edge effect." In the MM solution, we have not directly incorporated that physical characteristic. Nevertheless, this pulse expansion-point match MM solution has predicted that the line charge density is becoming very large near the edges.**

```
BarChart[α, AxesLabel → {"Basis #", "ρl (C/m)"},
 BarSpacing → -0.15, PlotLabel → "Microstrip in Infinite Dielectric"]
```



- Graphics -

- **The variation of C, Z0 and ϵreff are next observed as the number of basis functions is increased. This is called a "convergence study." It is not known a priori how many basis functions are needed in a MM solution to provide an accurate solution. A convergence study should show that the physical quantities are smoothly approaching an asymptote as the number of basis functions increases.**

```
ListPlot[Table[{numcells, CPUL}, {numcells, 50}], PlotJoined → True,
 AxesLabel → {"# basis", "C [F/m]"}, PlotLabel → "Infinite Dielectric",
 PlotRange → All, PlotStyle → RGBColor[1, 0, 0]]
```



- Graphics -

```
ListPlot[Table[{numcells, Z0}, {numcells, 50}], PlotJoined → True,
 AxesLabel → {"# basis", "Z0 [Ω]"}, PlotLabel → "Infinite Dielectric",
 PlotRange → All, PlotStyle → RGBColor[1, 0, 0]]
```



- Graphics -

```
ListPlot[Table[{numcells, ϵreff}, {numcells, 50}], PlotJoined → True,
 AxesLabel → {"# basis", "ϵreff"}, PlotLabel → "Infinite Dielectric",
 PlotRange → {0, ϵrbackgnd + 1}, PlotStyle → RGBColor[1, 0, 0]]
```



- Graphics -

- **Lastly, plot and list Z0 from the MM solution as a function of W/d and compare with the approximate solution given in (3.196) of Pozar, "Microwave Engineering." This approximate expression was presumably obtained by researchers curve fitting**

numerically accurate results, such as those from a MM solution like this one. These two solutions should be in close agreement only for ϵrbackgnd=1.

- **Approximate formula for the characteristic impedance of a quasi-static microstrip from (3.196) in Pozar.**

```
ϵe[Wod_, ϵr_] := (ϵr + 1) / 2 + (ϵr - 1) / 2 * 1 / Sqrt[1 + 12 / Wod]
Z0approx[Wod_, ϵr_] :=
 60 / Sqrt[ϵe[Wod, ϵr]] * Log[8 / Wod + Wod / 4]  /; Wod ≤ 1
Z0approx[Wod_, ϵr_] :=
 120 * Pi / (Sqrt[ϵe[Wod, ϵr]] * (Wod + 1.393 + 0.667 * Log[Wod + 1.444])) /;
  Wod > 1
```

```
numcells := 50
plot1 := ListPlot[Table[{Woverd, Z0}, {Woverd, 0.1, 10, 0.2}],
   AxesLabel → {"W/d", "Z0 [Ω]"}, PlotJoined → True,
   PlotLabel → "Moment Method/Infinite Dielectric",
   PlotStyle → RGBColor[1, 0, 0]]  ;
plot2 := ListPlot[Table[{Woverd, Z0approx[Woverd, ϵrbackgnd]},
    {Woverd, 0.1, 10, 0.2}], AxesLabel → {"W/d", "Z0approx [Ω]"},
   PlotJoined → True, PlotLabel → "Approximate Formula",
   PlotStyle → RGBColor[0, 0, 1]]  ;
Show[{plot1, plot2}, AxesLabel → {"W/d", "Z0 [Ω]"}, PlotLabel → "Both"]
```




Z0approx [Ω]
Approximate Formula
150
100
50
2
4
6
8
10
W/d


Z0 [Ω]
Both
150
100
50
2
4
6
8
10
W/d

- Graphics -

```
numcells := 50
TableForm[Table[
  {Woverd, Z0, Z0approx[Woverd, ϵrbackgnd]}, {Woverd, 0.2, 10, 0.4}],
 TableHeadings → {None, {"W/d", "Z0 [Ω]", "Z0approx [Ω]"}}]
```

W/d	Z0 [Ω]	Z0approx [Ω]
0.2	221.672	221.408
0.6	156.375	156.087
1.	126.82	126.613
1.4	108.104	108.016
1.8	94.7633	94.7707
2.2	84.6318	84.6128
2.6	76.6168	76.5471
3.	70.0886	69.9704
3.4	64.6522	64.4942
3.8	60.045	59.8562
4.2	56.0844	55.8728
4.6	52.6389	52.411
5.	49.6112	49.372
5.4	46.9275	46.6811
5.8	44.5309	44.2801
6.2	42.3765	42.1237
6.6	40.4284	40.1753
7.	38.6576	38.4057
7.4	37.0406	36.7908
7.8	35.5577	35.3107
8.2	34.1925	33.949
8.6	32.9312	32.6916
9.	31.7624	31.5268
9.4	30.6758	30.4446
9.8	29.6631	29.4363